

Федеральное государственное образовательное бюджетное учреждение высшего образования "Финансовый университет при Правительстве Российской Федерации"

Федеральное государственное автономное образовательное учреждение высшего образования "Московский физико-технический институт (государственный университет)"

Федеральное государственное бюджетное образовательное учреждение высшего образования "Российский экономический университет имени Г.В. Плеханова"

Федеральное государственное бюджетное образовательное учреждение высшего образования "Московский государственный технический университет имени Н.Э. Баумана (национальный исследовательский университет)"

Федеральное государственное бюджетное образовательное учреждение высшего образования "Московский государственный технологический университет "СТАНКИН"

Федеральное учебно-методическое объединение по укрупненной группе специальностей и направлений подготовки высшего образования "Информатика и вычислительная техника"

Фирма "1С"

19-я Международная научно-практическая конференция
"Новые информационные технологии в образовании.
(Использование технологий "1С" в образовании и их
применение для развития кадрового потенциала
цифровой экономики)

29 – 30 января 2019 года

Мастер-класс
**РАЗРАБОТКА РАСШИРЕНИЙ КОНФИГУРАЦИИ ДЛЯ
"1С:УНИВЕРСИТЕТ ПРОФ"**

Раздаточный материал

Москва, 2019

ПРАВО ТИРАЖИРОВАНИЯ ДОКУМЕНТАЦИИ ПРИНАДЛЕЖИТ ФИРМЕ "1С"

Получив настоящие материалы для обучения от фирмы "1С",
Вы тем самым даете согласие не
допускать их копирования без письменного
разрешения фирмы "1С"

© ООО "1С-Софт"

Москва, ул. Селезневская, д. 21

Телефон: (495) 737-92-57

Авторы методических материалов: Л.С. Волканин (lsv@usaaa.ru)

А.Ю. Хачай(1С@ta-ural.ru)

Оглавление	
Введение	4
1 Подготовка к работе	5
1.1 Установка информационных систем	5
1.2 Общие сведения о расширениях	5
1.3 Создание и подключение расширений	6
2 Сценарии использования.....	8
2.1 Изменение представлений объектов.....	8
2.2 Изменение интерфейса и подсистем	10
2.3 Устранение проблем совместной работы.....	11
2.4 Ограничение видимости персональных данных	13
2.5 Предотвращение появления дублей в справочниках	14
2.6 Доработка печатных форм.....	16
2.7 Добавление веб-сервиса.....	18
3 Примеры расширений для «1С:Университет ПРОФ».....	20
3.1 Учет договорных отношений средствами 1С:Университет ПРОФ.....	20
3.2 Военно-учетный стол	21
Заключение.....	22

Введение

Мастер-класс посвящен современному механизму адаптации прикладных решений под конкретного потребителя - механизму расширений.

Продукт масштаба «1С:Университет ПРОФ» при внедрении не обойдется без подстройки под особенности конкретного вуза. Вместе с новым функционалом приходят проблемы дальнейшего сопровождения, установки новых версий. Применение хорошо зарекомендовавших практик доработки решает проблему частично – как не старайся, а изменить типовой код без снятия конфигурации с поддержки нельзя, а это приведет к невозможности полностью автоматического обновления. Наиболее трудно сращивать изменения после серьезного рефакторинга кода со стороны поставщика. Если в вузе нет собственного специалиста, это может стать причиной отказа от доработок и существенного замедления внедрения. В мастер-классе рассматривается способ доработки функционала конфигурации без потери возможности автоматических обновлений, подходящий для всех современных решений 1С, включая используемые в образовательной деятельности:

- 1С:Университет и 1С:Университет ПРОФ,
- 1С:Колледж,
- 1С:Общеобразовательное учреждение,
- 1С:Управление учебным центром.

Мастер-класс будет интересен:

- сотрудникам управления информатизации вузов;
- сотрудникам, отвечающим за эксплуатацию решений 1С в вузе;
- всем желающим ознакомиться с возможностями расширения функционала «1С:Университет ПРОФ» при внедрении.

Рекомендуется знакомство с общими принципами разработки в 1С.

В рамках мастер-класса мы приводим только ключевые сведения о разработке расширений. Для более подробного знакомства рекомендуем обратиться к документации на платформу и к книге Е. Ю. Хрустальной «Расширения конфигураций. Адаптация прикладных решений с сохранением поддержки в облаках и на земле. Разработка в системе 1С:Предприятие 8.3. Издание 2» (ISBN978-5-9677-2834-1, М.: ООО "1С-Публишинг", 2018; артикул печатной книги по прайс-листу фирмы "1С": 4601546137920 или в информационной системе ИТС ПРОФ <https://its.1c.ru/db/pubextensions>).

1 Подготовка к работе

1.1 Установка информационных систем

Разработка расширений конфигурации для «1С:Университет ПРОФ» рассматривается в нижеследующем программном окружении:

- ОС Astra Linux Common Edition (2.12.8)
- СУБД PostgreSQL (9.6.6-1.1C)
- 1С:Предприятие 8.3 (8.3.12.1790)
- 1С:Университет ПРОФ, редакция 2.1 (2.1.4.3)

1.2 Общие сведения о расширениях

Расширения конфигурации были придуманы как механизм кастомизации (изменения для нужд конкретного пользователя), который бы позволял:

- легко обновлять кастомизированное решение на новую версию, избегая ручной работы по объединению конфигураций;
- включать кастомизацию при определенных условиях (например, если мы работаем в контексте определенной организации);
- снизить вероятность потери работоспособности кастомизации при переходе на новую версию исходной конфигурации;
- имел возможность отключения кастомизации в случае проблем для сохранения работоспособности приложения.

Расширение очень похоже на обычную конфигурацию. Оно так же представляется в виде дерева объектов. Для работы с расширением используются те же приёмы работы, что и с обычной конфигурацией. В отличие от конфигурации информационной базы, расширение не содержит и не должно содержать всю дополнительную функциональность.

С момента выхода в 2015 году механизм расширений подвергался переосмысливанию, доработкам и пересмотру функционала (табл. 1). Сейчас концепция механизма устоялась и её можно применять без особого страха и рисков.

Таблица 1.
Развитие возможностей в расширениях конфигураций

Основные возможности, доступные в расширениях конфигураций	Версия технологической платформы							
	8.3.6	8.3.7	8.3.8	8.3.9	8.3.10	8.3.11	8.3.12	8.3.13
Изменять управляемые формы, существующие в типовой конфигурации	+	+	+	+	+	+	+	+
Добавлять новые подсистемы, изменять состав имеющихся подсистем	+	+	+	+	+	+	+	+
Изменять имеющиеся роли, добавляя объекты, созданные в расширении	+	+	+	+	+	+	+	+
Изменять командный интерфейс (основного раздела, подсистем)	+	+	+	+	+	+	+	+
Добавлять новые отчёты и обработки	+	+	+	+	+	+	+	+
Создавать WS-ссылки, XDTO-пакеты, Web-сервисы и HTTP-сервисы		+	+	+	+	+	+	+

Основные возможности, доступные в расширениях конфигураций	Версия технологической платформы							
	8.3.6	8.3.7	8.3.8	8.3.9	8.3.10	8.3.11	8.3.12	8.3.13
Публиковать интернет-сервисы расширения на веб-сервере		+	+	+	+	+	+	+
Добавлять и изменять реквизиты и свойства заимствованных объектов, описывающих основные и вспомогательные формы			+	+	+	+	+	+
Заимствовать модули объектов, вызывать любые методы расширяемого модуля, добавлять обработчики событий				+	+	+	+	+
Создавать новые роли и изменять права доступа в заимствованных ролях				+	+	+	+	+
Изменять и расширять макеты табличных документов					+	+	+	+
Добавлять реквизиты и создавать собственные справочники, документы, планы обмена и регистры сведений						+	+	+
Формировать движения по регистрам для собственных и заимствованных документов							+	+
Создавать собственные перечисления							+	+
Использовать расширения в распределенной информационной базе							+	+
Создавать пользовательские стили и элементы стиля, заставку, логотип								+
Создавать планы видов характеристик, планы счетов, планы видов расчета, регистры накопления, бухгалтерии, расчета								+
Включать регистры любого вида в состав движений собственных и заимствованных документов расширения								+

1.3 Создание и подключение расширений

Процесс разработки расширения принципиально не отличается от обычного конфигурирования – расширение наполняется объектами конфигурации: заимствованными из конфигурации или созданными в расширении.

Чтобы редактировать имеющийся объект конфигурации, необходимо воспользоваться командой «Добавить в расширение» контекстного меню объекта конфигурации. После выполнения этой команды в расширении создается заимствованный объект. При этом автоматически заимствуются также все родительские объекты, если они еще не заимствованы в данном расширении. Заимствовать объекты нужно:


- когда они необходимы для визуального конструирования. Например, вы расширяете форму и добавляете реквизит формы типа СправочникСсылка.Валюты. Тогда вы должны заимствовать справочник Валюты, чтобы при подключении к типовой конфигурации быть уверенными в том, что в ней такой справочник всё ещё есть.

- когда они необходимы для работы кода. Например, в коде расширения вы обращаетесь к реквизиту документа «Учебный план» – Тип. Тогда этот реквизит также надо заимствовать, чтобы при подключении быть уверенным, что в типовой конфигурации у документа всё ещё существует такой реквизит.

При разработке расширений следует исходить из следующих предположений:

- расширение должно разрабатываться как автономный продукт, не опирающийся на наличие или отсутствие других расширений.
- одновременно может быть подключено более одного расширения, которое расширяет один и тот же объект расширяемой конфигурации.
- порядок исполнения программных модулей в результирующей конфигурации определяется порядком регистрации расширений в информационной базе расширяемой системы. Первым используется расширение, которое зарегистрировано последним.
- необработанное исключение, возникающее в любом из расширений (или расширяемой конфигурации) прерывает исполнение всей цепочки методов расширений и распространяется в расширяемой конфигурации.

Вы можете выгрузить расширение в файл (расширением *.cfe), и загрузить из файла. Форма управления расширениями конфигурации открывается по одноименной команде меню «Все функции» (Рис 1).



Расширение	Версия	Назначение	Активно	Безопасный режим, имя профиля	Защита от опасных действий	Используется в распределенной ИБ	Область действия
Решение проблемных ситуаций	201807051700	Исправление	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Приемная комиссия	2018.07.12	Адаптация	✓	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	Разделение данных
Договоры (график платежей)	2018.08.20	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Скрытие ПДн	2018.06.07	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Учебный план (представление)	2017.10.23	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Физические лица	2018.07.01	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Контрагенты	2018.07.01	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Приказы	2018.09.10	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Редактирование значений хара...	2018.06.07	Адаптация	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных
Договоры (печать)	2018.08.25	Адаптация	✓	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	Разделение данных
Воинский учет	1.0.0.2	Дополнение	✓	<input type="checkbox"/>	✓	<input type="checkbox"/>	Разделение данных

Рис 1 Управление расширениями конфигурации

2 Сценарии использования

Приведём примеры встречающихся практически во всех вузах доработок, которые можно реализовать с использованием расширений конфигураций.

2.1 Изменение представлений объектов

При контроле правильности настроек программы, заполнения приказов и других документов удобно использовать короткие и информативные представления объектов информационной системы. Согласитесь, в заявлении абитуриента быстрее прочитывается «ПК (магистратура) 2018 г.» вместо «Приемная кампания 000000004 от 01.06.2018 0:00:00».

Обычно в 1С ссылки на элементы справочников отображаются в виде кода или наименования (как задано в конфигураторе), а для документа основное представление – тип документа, номер, дата и время. Изменить стандартное поведение разработчику позволяет процедура (обработчик события) платформы `ОбработкаПолученияПредставления`.

Предположим, что мы выполнили анализ типовых ошибок при подготовке приказов (выбираются учебные планы вместо рабочих, не тот профиль и т.д.), и решили отображать в приказах поле с планом в виде «РП 2018 г. 07.03.01/АП очная №335 от 15.06.2015». Создадим новое расширение, которое будет использовать такую возможность:

- заимствовать модули объектов, вызывать любые методы расширяемого модуля, добавлять обработчики событий (режим совместимости с версией 8.3.9).

Для формирования строки в нужном формате потребуется заимствовать в расширение следующие объекты основной конфигурации:

- документ «Учебный план» и реквизиты Тип, Специальность, УчебныйГод, ФормаОбучения;
- предопределенные значения справочника «Типы учебных планов».

При заимствовании реквизитов в расширении автоматически появились соответствующие справочники. Желательно также добавить в расширение и поставить на контроль реквизиты, которые мы будем использовать:

- справочник «Специальности» – реквизит КодСпециальности;
- справочник «Учебные годы» – реквизит НачальныйГод;
- справочник «Форма обучения» – не требуется заимствование реквизитов.

Реализуем в модуле менеджера документа «Учебный план» формирование строки представления в требуемом формате (Листинг 1) и проверим работу алгоритма, открыв любой приказ (Рис 2). Обработчики событий в расширении рекомендуют определять с аннотациями `&Перед` и `&После`, чтобы расширяемый метод выполнялся в любом случае.

Листинг 1 Процедуры менеджера документа «Учебный план»

```
&После ("ОбработкаПолученияПолейПредставления")
Процедура Расш1_ОбработкаПолученияПолейПредставления (Поля,
СтандартнаяОбработка)
    Поля.Добавить ("Специальность");
    Поля.Добавить ("УчебныйГод");
    Поля.Добавить ("ФормаОбучения");
```


КонецПроцедуры

```
&После ("ОбработкаПолученияПредставления")
Процедура Расш1_ОбработкаПолученияПредставления(Данные, Представление,
СтандартнаяОбработка)
    СтандартнаяОбработка = Ложь;
    Если Данные.Тип = Справочники.ТипыУчебныхПланов.УчебныйПлан Тогда
        Префикс = "УП";
    ИначеЕсли Данные.Тип = Справочники.ТипыУчебныхПланов.РабочийПлан
    Тогда
        Префикс = "РП";
    КонецЕсли;
    Представление =
СтрковыеФункцииКлиентСервер.ПодставитьПараметрыВСтроку (
    "%1 %2 %3 %4 №%5 от %6",
    Префикс,
    Формат (Данные.УчебныйГод.НачальныйГод, "ДФ='уууу ' 'г.' ' '"),
    Данные.Специальность.КодСпециальности,
    НРег (Данные.ФормаОбучения) ,
    ПрефиксацияОбъектовКлиентСервер.НомерНаПечать (Данные.Номер) ,
    Формат (Данные.Дата, "ДФ=dd.ММ.уууу")
);
КонецПроцедуры
```

N	Вид приказа	N	В выпуску	Физическое лицо	Зачетна...	Дата начала	Учебный план
1	Зачисление в вуз	1	<input type="checkbox"/>	Арзумян Гарик Владимирович	285811	15.08.2014 0:00:00	РП 2014 г. 07.03.01 очная №140 от 02.06.2014
		2	<input type="checkbox"/>	Васильченко Николай Игоревич	577976	15.08.2014 0:00:00	РП 2014 г. 07.03.01 очная №140 от 02.06.2014
		3	<input type="checkbox"/>	Долженко Иван Алексеевич	665527	15.08.2014 0:00:00	РП 2014 г. 07.03.01 очная №140 от 02.06.2014
		4	<input type="checkbox"/>	Слепенко Елена Владимировна	395822	15.08.2014 0:00:00	РП 2014 г. 07.03.01 очная №140 от 02.06.2014

Рис 2 Представление ссылки на учебный план в приказе

Отметим особенности и ограничения формирования представления:

- Для формирования представления доступны не все поля объекта, а только перечисленные в процедуре `ОбработкаПолученияПолейПредставления`;
- Обращение к реквизитам через точку (`Специальность.КодСпециальности`) приведёт к считыванию и кешированию всех полей и табличных частей, для сложных справочников лучше выбрать нужные поля запросом.
- В некоторых случаях необходимые для формирования представления поля содержатся в табличной части (напр., профиль учебного плана) и недоступны в параметре `Данные`. Можно добавить реквизит `Ссылка` в массив необходимых полей и извлечь данные из базы запросом.

Опытные пользователи и разработчики, знакомые с «ИС:Университет ПРОФ», отметят, что предложенное решение не везде поменяет формат ссылок на учебные планы. Действительно, в некоторых отчетах, договорах на обучение, ссылка будет выглядеть примерно так – «Очная, факультет Архитектуры, Архитектура, Бакалавр, 2018 – 2019». В них используется не сформированное платформой представление, а заполняемый при сохранении учебного или рабочего плана реквизит `ПредставлениеУчебногоПлана`.

При необходимости, можно заимствовать модуль `УчебныйПланСервер` и переопределить в нём функцию `ПолучитьПредставлениеУчебногоПлана(УчебныйПлан)`. В примере (Листинг 2) используется аннотация `&Вместо`, т.к. `&Перед` и `&После` для функций недоступны. Если нужен результат работы оригинальной функции, используйте специальную конструкцию `ПродолжитьВызов(...)` с исходными параметрами.

Листинг 2 Переопределение функции общего модуля УчебныйПланСервер

```
&Вместо ("ПолучитьПредставлениеУчебногоПлана")  
Функция Расш1_ПолучитьПредставлениеУчебногоПлана (УчебныйПлан)  
    Результат = ПродолжитьВызов (УчебныйПлан) ;  
    Возврат Результат ;  
КонецФункции
```

Для заполнения реквизита по новому алгоритму все учебные планы нужно будет однократно перепровести.

2.2 Изменение интерфейса и подсистем

Кафедры и деканаты часто жалуются на слишком большое количество элементов интерфейса. Действительно, пользователям лучше показывать ссылки только на используемые ими объекты – справочники, приказы, отчеты. Нужного результата можно достичь несколькими способами и без использования механизма расширений:

- настроить профили групп доступа исходя из выполняемых трудовых функций, если пользователю не дан ролевой доступ к какому-то объекту конфигурации, в интерфейсе тот показан не будет;
- при использовании интерфейса «Такси» вручную убрать из навигации лишние ссылки, нажав на «шестерёнку» (Рис 3), и скопировать настройки пользователям;




Рис 3 Возможность настройки интерфейса стандартными средствами

Когда использование расширений оправдано? В том случае, если при построении интерфейса необходимо переопределить логику разработчиков. Перечисленные выше способы не позволят в подсистему «Управление студенческим составом» добавить отчёт «Задолжники по договорам».

Предположим, в «1С:Университет ПРОФ» работают сотрудники бухгалтерии, ответственные за взаиморасчеты со студентами, и нужно организовать простой интерфейс. Создадим новое расширение, которое будет использовать такие возможности:

- добавлять новые подсистемы, изменять состав имеющихся подсистем;
- изменять имеющиеся роли, добавляя объекты, созданные в расширении;
- изменять командный интерфейс (основного раздела, подсистем);
- добавлять новые отчёты и обработки.

Продемонстрируем подход к решению поставленной задачи (Рис 4):

- создадим структуру подсистем, исходя из направлений деятельности вуза;

- заимствуем из основной конфигурации и включим в новые подсистемы необходимые бухгалтерии объекты – справочники, документы, отчеты.
- добавим новую команду «Приказы на стипендию», открывающую список приказов с заранее настроенным отбором;
- добавим новый отчет, который нужен бухгалтерии;
- добавим роль, дающую доступ к новым объектам.




Рис 4 Структура расширения для организации интерфейса

Убедимся, что в пользовательском режиме появилась новая подсистема с требуемыми ссылками (Рис 5).

Платное обучение	Стипендии	Отчеты
Приказы	Приказы на стипендию	Задолжники по договорам Списки получающих стипендию
Проживание	См. также	
Приказы на заселение и выселение	Физические лица	

Рис 5 Интерфейс расширения в пользовательском режиме

Таким же образом можно создать интерфейсы для других типов подразделений – деканатов, кафедр, военно-учетного стола и т.п.

2.3 Устранение проблем совместной работы

Распространена ситуация, когда пользователь закрывает любую экранную форму нажатием на самую большую кнопку «Записать и закрыть», «Провести и закрыть».




Рис 6 Активизированная по умолчанию кнопка документа

Если документ (Рис 6) только посмотрели и ничего не поменяли, команда «Провести и закрыть» запустит нежелательную цепочку действий – перезаполнение

регистров, регистрация изменений для обмена и т.п. Неприятнее другая ситуация – один пользователь редактирует данные документа или справочника, а другой в это время открыл тот же объект, посмотрел и закрыл командой «Записать и закрыть». Первый увидит ошибку «Данные были изменены или удалены другим пользователем», не сможет сохранить изменения и будет долго ругаться. Сам того не желая, пользователь организовал ситуацию одновременного редактирования, которая должна специальным образом обрабатываться как платформой, так и разработчиками решения.

Опытные пользователи скажут, и будут правы, что при попытке одновременного редактирования одного и того же элемента справочника платформа сама заблокирует объект и выдаст сообщение об ошибке непосредственно в момент начала изменений. К сожалению, этот механизм не срабатывает при изменении поля ввода, которое непосредственно не связано с реквизитами редактируемого объекта. В реальных внедрениях встречаются, например, такие проблемы одновременного редактирования:

- не блокируется элемент справочника «Физические лица», если в форме студента редактируются контактные данные и другие подчиненные данные;
- не блокируется при редактировании документ «Приказ»;
- список легко можно продолжить.

Первая часть предлагаемого решения проблемы – не выполнять стандартные операции, если данные на форме не менялись. Создадим новое расширение и заимствуем из формы документа «Учебный план» обработчик события ПередЗаписью (Рис 7), реализуем алгоритм доработки (Листинг 3).

Листинг 3 Проверка модифицированности перед записью объекта

```
&НаКлиенте  
Процедура Расш3_ПередЗаписьюПеред(Отказ, ПараметрыЗаписи)  
    Если Не ЭтаФорма.Модифицированность Тогда  
        Отказ = Истина;  
        ЭтаФорма.Закреть ();  
    КонецЕсли;  
КонецПроцедуры
```




Рис 7 Добавление расширения обработчика события ПередЗаписью

С одной стороны, подобные изменения необходимо внести во многие экранные формы. С другой стороны, велика вероятность, что будущие изменения основной конфигурации не отразятся на работоспособности расширения.

Вторая часть предлагаемого решения – программно заблокировать объект при начале редактирования подчиненных данных, для этого можно использовать метод управляемой формы `ЗаблокироватьДанныеФормыДляРедактирования()`. Форма элемента справочника «Физические лица» содержит 27 динамических списков, для примера рассмотрим раздел «Контактная информация» (Рис 8). В расширение конфигурации заимствовать нужно форму записи регистра `КонтактнаяИнформация`




Рис 8 Исходные данные для списка «Контакты» физического лица

Владельцем открытой формы редактирования контактных данных является таблица «Контакты» формы физического лица, данные которого нужно заблокировать для редактирования. Блокировку можно выполнить (Листинг 4) как после открытия контактной информации, так и при изменении любого из реквизитов.

Листинг 4 Блокировка объекта для изменения при начале редактирования

```
&НаКлиенте
Процедура Расш3_ПриОткрытииПосле(Отказ)
    УправляемаяФорма = ВладелецФормы;
    Пока ТипЗнч(УправляемаяФорма) <> Тип("УправляемаяФорма") Цикл
        УправляемаяФорма = УправляемаяФорма.Родитель;
    КонечЦикла;
    УправляемаяФорма.ЗаблокироватьДанныеФормыДляРедактирования();
КонечПроцедуры
```

Устранение проблем совместной работы – отдельное направление в развитии прикладного решения, часто требующее глубокого анализа и существенных изменений в программном коде. Причины некоторых проблем можно самостоятельно устранить в расширении конфигурации.

2.4 Ограничение видимости персональных данных

Необходимость изменения форм ввода данных возникает в разных случаях. С помощью расширения можно доработать внешний вид формы, добавить в нее недостающие данные, дополнить ее командный интерфейс. Если же доработать модуль формы, можно не добавить, а наоборот, удалить или скрыть часть полей ввода.

В зависимости от должностных обязанностей пользователя в карточке физического лица можно не показывать часть вкладок и элементов формы. Например, ответственному за движение контингента не всегда нужно видеть паспортные данные, а информацию о военной службе можно показывать только работникам военно-учетного стола.

Создадим новое расширение и продемонстрируем подход к решению задачи:

- заимствуем справочник «Физические лица» и форму элемента;

- создадим роли, исходя из выполняемых пользователями обязанностей (Рис 9);
- реализуем ограничение видимости при создании формы (Листинг 5);
- в пользовательском режиме в разделе «Администрирование», «Настройки пользователей и прав» создадим профиль групп доступа (Рис 10).




Рис 9 Структура расширения для организации ролевой видимости данных




Рис 10 Создание профиля доступа с использованием роли из расширения

Листинг 5 Ограничение видимости элементов формы на основе ролей

```
&НаСервере
Процедура Расш4_ПриСозданииНаСервереПосле (Отказ, СтандартнаяОбработка)
```

```
    Если НЕ Пользователи.РолиДоступны("Расш4_ВУС") Тогда
        Элементы.ВоеннаяСлужба.Видимость = Ложь;
    КонецЕсли;
```

```
КонецПроцедуры
```

Предлагаемое решение позволяет организовать динамическое управление видимостью элементов формы, и не позволяет пользователю включить самостоятельно скрытые элементы с использованием команды «Изменить форму».

2.5 Предотвращение появления дублей в справочниках

Мы уже говорили, что причинами многих проблем, возникающих при эксплуатации системы, является неверный ввод исходных данных пользователями. Например, операторы приёмной комиссии заполняют десятками дублирующих записей справочник «Контрагенты», делают ошибки и опечатки в наименованиях. Полезно будет перед сохранением нового элемента справочника проверить на существования подобных.

Для поиска и удаления дублей удобно использовать одноименную подсистему библиотеки стандартных подсистем, ключевым преимуществом которой является поиск по подобию (нечёткий поиск, fuzzy search).

Создадим новое расширение и продемонстрируем подход к решению задачи:

- заимствуем общий модуль ПоискИУдалениеДублейПереопределяемый и добавим справочник Контрагенты в метод ПриОпределенииОбъектовСПоискомДублей();
- заимствуем справочник Контрагенты и форму элемента,
- в менеджере справочника опишем параметры поиска дублей – совпадение по типу и подобию наименования, а если необходимо, то и реализуем свой собственный алгоритм сравнения при поиске дублей (Листинг 7);
- в форме элемента добавим обработчик события «Перед записью», в нём обратимся к методу БСП НайтиДублиЭлемента(), и просто выведем полученный список возможных дублей на экран (Листинг 6)

Листинг 6 Поиск дублей, перед записью элемента справочника «Контрагенты»

```
&НаКлиенте
Процедура Расш5_ПередЗаписьюПосле (Отказ, ПараметрыЗаписи)
    СписокДублей = Расш5_НайтиДублиЭлемента (Объект) ;
    Если СписокДублей.Количество () > 0 Тогда
        Отказ = Истина;
        СписокДублей.ПоказатьВыборЭлемента ("Похожие элементы") ;
    КонецЕсли;
КонецПроцедуры

&НаСервереБезКонтекста
Функция Расш5_НайтиДублиЭлемента (Знач Проверяемый)
    Результат = Новый СписокЗначений;
    ЭталонныйОбъект =
ДанныеФормыВЗначение (Проверяемый, Тип ("СправочникОбъект.Контрагенты") ) ;
    ВозможныеДубли = ПоискИУдалениеДублей.НайтиДублиЭлемента (
        "Справочник.Контрагенты", ЭталонныйОбъект, Неопределено ) ;
    Для Каждого Строка Из ВозможныеДубли Цикл
        Если ЭталонныйОбъект.Ссылка <> Строка.Ссылка Тогда
            Результат.Добавить (Строка.Ссылка) ;
        КонецЕсли;
    КонецЦикла;
    Возврат Результат;
КонецФункции
```

Листинг 7 Настройки поиска в модуле менеджера справочника «Контрагенты»

```
Процедура ПараметрыПоискаДублей (ПрикладныеПараметры,
ДополнительныеПараметры) Экспорт
    ДобавитьПравило (ПрикладныеПараметры, "Равно", "ТипКонтрагента") ;
    ДобавитьПравило (ПрикладныеПараметры, "Подобно", "Наименование") ;
КонецПроцедуры

Процедура ДобавитьПравило (ПрикладныеПараметры, Правило, Реквизит)
    Отбор = Новый Структура ("Правило, Реквизит", Правило, Реквизит) ;
    Строки = ПрикладныеПараметры.ПравилаПоиска.НайтиСтроки (Отбор) ;
    Если Строки.Количество () = 0 Тогда
        НовоеПравило = ПрикладныеПараметры.ПравилаПоиска.Добавить () ;
        НовоеПравило.Правило = Правило;
        НовоеПравило.Реквизит = Реквизит;
    КонецЕсли;
```


КонецПроцедуры

```
Процедура ПриПоискеДублей(ТаблицаКандидатов, ДопПараметры) Экспорт
    Для Каждого Строка Из ТаблицаКандидатов Цикл
        Строка.ЭтоДубли = Истина; // анализ :)
    КонецЦикла;
КонецПроцедуры
```

ГОУ ВПО Ставропольский государственный педагогический институт (Контрагенты)

The screenshot shows a web application interface for finding contractors. At the top, there are navigation tabs: 'Основное', 'Ведомственная принадлежность', 'Контактная информация', and 'Еще...'. Below these, there are two buttons: 'Юридическое лицо' (highlighted) and 'Физическое лицо'. A search field labeled 'Наименование:' contains the text 'ГОУ ВПО Ставропольский государственный педагогический институт'. A dialog box titled 'Похожие элементы' is open, showing a list of two identical entries: 'ГОУ ВПО Ставропольский государственный педагогический институт'. The dialog box has 'OK' and 'Отмена' buttons.

Рис 11 Найденные дубли контрагента

Проверим наше решение – создадим новый элемент с опечаткой в наименовании (институт) и попробуем его сохранить, будут показаны найденные дубликаты (Рис 11).

2.6 Доработка печатных форм

Изменения в печатных формах – обязательный этап внедрения «1С:Университет». Заказчиками доработок обычно являются приемная комиссия и подразделения, ответственные за подготовку приказов и справок. Изменения кода конфигурации в ряде случаев можно избежать:

- если печатная форма реализована с использованием БСП, её может самостоятельно редактировать ответственный пользователь (Рис 12);
- если печатная форма реализуется как внешняя обработка;
- если настроена бесшовная интеграция с системой электронного документооборота «1С:Документооборот», в которой подготовлены шаблоны документов.




Рис 12 Самостоятельное редактирование пользователем макета печатной формы

Если перечисленные варианты не подходят, или требуется не разработка нового макета, а изменение уже существующего, можно воспользоваться возможностями расширений. Наши коллеги, например, отмечают, что при всей гибкости настройки печати приказов «1С:Университет» не даёт возможности глобально поменять шапку документов.

Создадим новое расширение, которое будет использовать такой функционал:

- изменять и расширять макеты табличных документов (появилось в версии 8.3.10)

Из основной конфигурации заимствуем общий макет ШапкаВуза и настроим печатную форму (Рис 13).




Рис 13 Общий макет в расширении конфигурации

Обратим внимание на особенность заимствования макетов табличных документов, можно настраивать способ расширения – объединение или замещение (Рис 14). В первом случае (это настройка по умолчанию) построение итогового документа будет происходить немного медленнее.




Рис 14 Особенности расширения макетов табличных документов

Команды печати могут быть добавлены в расширение различными способами. Полноценную реализацию печатной формы в этом мастер-классе мы не рассматриваем.

2.7 Добавление веб-сервиса

Информирование абитуриентов о плане набора и количестве поданных заявлений обычно делается через сайт вуза. Однако, реализация на стороне сервера 1С REST-интерфейса и использование веб-приложений на основе Angular, React и пр., позволяет организовать информационную панель вообще без изменения серверного кода сайта вуза и настройки обменов данными. Создадим новое расширение, использующее возможность:

- создавать WS-ссылки, XDTO-пакеты, Web-сервисы и HTTP-сервисы;
- публиковать интернет-сервисы расширения на веб-сервере.




Рис 15 Структура веб-сервиса в расширении

Создадим в расширении веб-сервис, в качестве примера продемонстрируем получение и публикацию в формате JSON плана набора (Рис 15, Листинг 8).

Листинг 8 Получение и публикация плана набора через веб-сервис

```

Функция ШаблонПланНабораПолучить (Запрос)
    Ответ = Новый HTTPСервисОтвет (200);
    Ответ.Заголовки.Вставить ("Content-type", "application/json");
    Запрос = Новый Запрос;
    Запрос.УстановитьПараметр ("ГодПоступления", '20180101000000');
    Запрос.Текст =
        "ВЫБРАТЬ
        |     Планы.КонкурснаяГруппа.Представление КАК КонкурснаяГруппа,
        |     Планы.ОснованиеПоступления.Представление КАК Основание,
        |     Планы.УровеньБюджета.Представление КАК УровеньБюджета,
        |     Планы.КоличествоМест КАК Мест
        | ИЗ
        |     РегистрСведений.ПланыНабора.СрезПоследних(, ) КАК Планы
        | ГДЕ
        |     Планы.ПриемнаяКампания.ГодПоступления = &ГодПоступления";
    Выб = Запрос.Выполнить().Выбрать();
    мПланНабора = Новый Массив;
    Пока Выб.Следующий() Цикл
        мПланНабора.Добавить (Новый
        Структура ("КонкурснаяГруппа, Основание, УровеньБюджета, КоличествоМест",
        Выб.КонкурснаяГруппа, Выб.Основание, Выб.УровеньБюджета, Выб.Мест));
        КонецЦикла;
    JSON = Новый ЗаписьJSON;
    JSON.УстановитьСтроку(); ЗаписатьJSON (JSON, мПланНабора);
    Ответ.УстановитьТелоИзСтроки (JSON.Закреть());
    Возврат Ответ;
КонецФункции
    
```




Рис 16 Настройка публикации веб- и HTTP-сервисов расширений

Выберем в конфигураторе команду «Публикация на веб-сервере» в меню Администрирование и проверим, что HTTP-сервисы расширений публикуются (Рис 16). Затем проверим, что при обращении к веб-сервису данные возвращаются в нужном виде:

```
$ curl http://ws@server1C/uni/hs/priem/plan/
```

```
[
  {
    "КонкурснаяГруппа": "Архитектура_Очная_Бюджет",
    "Основание": "бюджетная основа",
    "УровеньБюджета": "Федеральный бюджет",
    "Мест": 56
  },
  ... <текст обрезан> ...
]
```

3 Примеры расширений для «1С:Университет ПРОФ»

В этом разделе приведены примеры реально используемых вузами расширений.

3.1 Учет договорных отношений средствами 1С:Университет ПРОФ

Федеральный закон "Об образовании в Российской Федерации", постановление Правительства РФ от 15 августа 2013 г. № 706 "Об утверждении Правил оказания платных образовательных услуг" указывают, что договор об оказании платных образовательных услуг должен содержать полную стоимость этих услуг. С учетом допустимой индексации и возможных переводов обучающегося на другие образовательные программы с иной стоимостью обучения полная стоимость договора может меняться и должна фиксироваться дополнительными соглашениями к договору. Вузу необходим функционал автоматического расчета суммы договора как при создании базового договора, так и при работе с дополнительными соглашениями, изменяющими срок и/или стоимость обучения.




Рис 17 Схема взаимодействия договоров и дополнительных соглашений

В статье Слесаревой Э.В «Учет договорных отношений средствами "1С:Университет ПРОФ"»¹ описано создание расширения конфигурации, с такими изменениями и дополнениями типового функционала:

- автоматический расчет полной стоимости обучения и графика платежей по периодам, если в качестве способа расчета указан добавленный вузом способ;
- автоматический расчет полной стоимости обучения при создании ДС к договору на изменение ООП (при переводе студента) или индексацию;
- автоматический расчет сроков договора в случае ДС на изменение срока обучения (академический отпуск и др.), изменение ООП или расторжения договора;
- формирование печатной формы графика платежей с учетом всех дополнительных соглашений к договору.

Перечисленные доработки позволили гибко настроить макеты договоров с выводом необходимых данных в печатную форму договора, что положительно повлияло на оперативность создания договоров и дополнительных соглашений к ним.

¹ <https://educonf.1c.ru/conf2018/thesis/2610/>

3.2 Военно-учетный стол

Практически в любом вузе постановка студентов на воинский учет ведется силами созданных военно-учетных столов, в функции которых входит в том числе документальное оформление сведений о гражданах, состоящих на воинском учете. Ведение личных дел, формирование отчетов реализовано в расширении конфигурации (Рис 17).




Рис 18 Интерфейс расширения «Военно-учетный стол»

В расширении реализованы также регламентированные печатные формы:

- листок сообщения об изменении сведений о гражданах, состоящих на воинском учете;
- сведения о гражданине, подлежащем воинскому учету, принятии (поступлении) его на работу (учебу в образовательное учреждение) или увольнении (отчислении) его с работы (из образовательного учреждения);
- сопроводительное письмо;
- список граждан, пребывающих в запасе, обучающихся и работающих.

Так как большинство документов воинского учета граждан, оформляемых в вузе, требует регистрации, подписания руководителем, обеспечения хранения на длительный срок, в расширении реализована поддержка систем электронного документооборота. Рекомендуется интеграция с решением «1С:Документооборот»², но если по каким-то причинам используется другая система документооборота, то в расширении продублирован функционал нумераторов.

Расширение проходит апробацию в Саратовской государственной юридической академии. Авторы благодарят начальника 2-го отдела Макарова Леонида Александровича за методическую поддержку.

² Подробнее см. – Практические сценарии бесшовной интеграции "1С:Университет" и "1С:Документооборот" при автоматизации бизнес-процессов вуза. / Волканин Л. С.; Хачай А. Ю., Информатика и образование. 2018; № 3 (292). стр. 39-43

Заключение

Каждый вуз уникален. Не нужно требовать от разработчиков «1С:Университет» реализацию всех мыслимых и немыслимых особенностей учета. Необходимо обеспечить современную инфраструктуру: в первую очередь поддержку последней версии технологической платформы и стандартных библиотек (БСП, БИД и др).

Рекомендуем функциональные особенности, отражающие многолетнюю практику вуза управления службами и контингентом, помещать в подключаемые к конфигурации расширения.

Об авторах

Волканин Леонид Сергеевич, к.ф.-м.н., начальник отдела ИТ Уральского государственного архитектурно-художественного университета, 1С:Эксперт по технологическим вопросам, г. Екатеринбург; e-mail: lsv@usaaa.ru

Хачай Андрей Юрьевич, к.ф.-м.н., доцент кафедры вычислительной математики и компьютерных наук Уральского федерального университета, директор ООО "Технологии автоматизации", г. Екатеринбург; тел. (343) 382-13-81, e-mail: 1c@ta-ural.ru

ООО "Технологии автоматизации" – официальный партнер Фирмы "1С", 1С:Центр компетенции по образованию, 1С:Центр компетенции по документообороту, 1С:Центр сертифицированного обучения, 1С:Авторизованный центр сертификации.

Адрес: 620014, г. Екатеринбург, ул. Радищева, 6а, "Суворов", оф. 304

Тел. (343) 247-85-45, e-mail: 1c@ta-ural.ru, <http://www.ta-ural.ru>